
Cleaning Up VSAM with
REXX/VSE

I recently came across the following 
question on the Internet: Is there an

easy way to delete all of the VSAM clusters
out of a catalog? While there are a number
of easy ways to do this, I thought I'd provide
a REXX/VSE solution by creating a couple
of routines that would do the job. The first,
REXXDELC, is a REXX program that will
accept the name of a file and a catalog and
call IDCAMS, telling the VSAM utility to
delete the specified cluster from a specific
catalog. The second program, REXXCLUS,
is a REXX program that will accept the
name of a catalog and pass a "LISTCAT
CLUSTER" request to IDCAMS. It will
then interrogate the response, getting a list
of all of the clusters in the catalog. Having
done so, it will invoke REXXDELC, passing
control to another REXX program to do the
actual deleting of the file.

Of course, you must be extremely careful
when using this procedure. Granted, you
may want to limit specific catalog names
and file names so that someone doesn't
accidentally delete something vital! This is
a good working example of how a little bit
of code can provide you with a great deal of
power. Figure 1 provides the code to delete
a VSAM cluster.

Keep in mind that you can just delete a
single specific cluster by running the JCL
shown in Figure 2. In this way, you can make
the job parameter driven, allowing more
dynamic use of a common IBM utility.

This brings us to the engine behind the
function, REXXCLUS. The program pre-
sented in Figure 3 will gather the names of
the clusters and call REXXDELC to delete
them one at a time. It will also provide a
nice condensed listing of all of the results
of your request. Figure 4 presents the JCL
necessary to perform this mass delete of
clusters from a single catalog.

/*
This program will delete a cluster from the VSAM catalog

Get the passed parms and validate them
*/

ARG filename catname
x = length(catname)
if x = 0 then do

say 'Missing catalog fileid'
exit 8
end

x = length(filename)
if x = 0 then do

say 'Missing cluster fileid'
exit 8
end

/*
Build the IDCAMS parms to pass and call IDCAMS
*/

idcams_input.0 = 2    /* Two "cards" to pass  */
idcams_input.1 = 'DELETE' filename 'PURGE -'
idcams_input.2 = '        CATALOG(' catname ')'
CALL OUTTRAP idcams_output.
CALL REXXIPT idcams_input.
ADDRESS LINK 'IDCAMS MARGINS(1 80)'
CRC = rc

/*
Display on SYSLST to indicate if the delete worked or not
*/
say ' '
do line = 4 to idcams_output.0

x=substr(idcams_output.line,1,8)
if x = 'IDC3012I' then say idcams_output.line
if x = 'IDC0550I' then say idcams_output.line
if CRC > 8 then say idcams_output.line
end

exit CRC

FIGURE 1: CODE TO DELETE A VSAM CLUSTER

// JOB DELETE A CLUSTER
// LIBDEF *,SEARCH=lib.sublib
// SETPARM CATID='VSESP.USER.CATALOG'
// SETPARM FILEID='FILE.NAME.GOES.HERE'
// EXEC REXX=REXXDELC,PARM='&FILEID &CATID'
/&

FIGURE 2: CODE TO DELETE A SPECIFIC CLUSTER

Enjoy, and please be careful!  

NaSPA member Leo J. Langevin has been a VSE systems programmer
since VSE was introduced to the world. He is currently a senior systems
programmer with Connectivity Systems, writing and supporting a
large range of TCP/IP-based products. He can be reached at
leo@tcpip4vse.com.

/*

This program will ask IDCAMS for a listing of cluster names
and then ask to delete each and every one of them.

*/

ARG catname /* Get the catalog name */

x = length(catname) /* Was data passed ? */
if x = 0 then do

say 'Missing catalog fileid'
exit 8
end

CALL OUTTRAP idcams_output. /* Setup a trap for the output */
CALL REXXIPT idcams_input. /* Setup a trap for the input */

idcams_input.0 = 1 /* Only one "card" */
idcams_input.1 = 'LISTCAT CL CAT('||catname||')'

ADDRESS LINK 'IDCAMS MARGINS(1 80)'

CRC = rc
if rc = 0 then do

cluster.0 = 0
call gather_names
call delete_clusters
end

else
say 'IDCAMS setup request failed. RC='rc

exit CRC

gather_names:
do line = 1 to idcams_output.0
x = substr(idcams_output.line,1,10)
if x = 'CLUSTER --' then do

x = subword(idcams_output.line,3,1)
cluster.0 = cluster.0 + 1
n = cluster.0
cluster.n = x
end

end
return

delete_clusters:
do count = 1 to cluster.0

call REXXDELC cluster.count catname
CRC = rc
end

return

FIGURE 3: PROGRAM TO OBTAIN A LISTING OF 
CLUSTER NAMES FOR DELETION

// JOB DELETE EVERY CLUSTER!
// LIBDEF *,SEARCH=lib.sublib
// EXEC REXX=REXXCLUS,PARM='USER.TEST.CATALOG'
/&

FIGURE 4: JCL NECESSARY TO PERFORM 
A MASS DELETE OF CLUSTERS


